# Contributor: Vasco Costa <vasco dot costa at geekslot dot com>
# Maintainer: Evangelos Foutras <evangelos@foutrelis.com>

pkgname=pbzip2
pkgver=1.1.13
pkgrel=3
pkgdesc="Parallel implementation of the bzip2 block-sorting file compressor"
arch=('loong64' 'x86_64')
url="http://compression.ca/pbzip2/"
license=('BSD')
depends=('bzip2' 'gcc-libs')
source=(https://launchpad.net/$pkgname/1.1/$pkgver/+download/$pkgname-$pkgver.tar.gz{,.asc})
sha256sums=('8fd13eaaa266f7ee91f85c1ea97c86d9c9cc985969db9059cdebcb1e1b7bdbe6'
            'SKIP')
validpgpkeys=('F3D99FF0F0AA96A659AD61A9D332F1B09A0D9836')

prepare() {
  cd "$srcdir/$pkgname-$pkgver"

  # Use our CXXFLAGS and LDFLAGS
  sed -i '/^\(CXX\|LD\)FLAGS =/d' Makefile
}

build() {
  cd "$srcdir/$pkgname-$pkgver"
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make PREFIX=/usr DESTDIR="$pkgdir" install
  install -Dm644 COPYING "$pkgdir/usr/share/licenses/$pkgname/COPYING"
}

# vim:set ts=2 sw=2 et:
